


GIRASOLE SCOMPOSTO

LINEE CURVE E COLORI

Oltre le linee dritte, oltre i cerchi, i semicerchi e gli archi.
Creare un disegno con sole linee curve.



00. Solo FORWARD e RIGHT

Questo progetto disegna un girasole.

È un progetto particolare perché vengono utilizzate due sole istruzioni, FORWARD e RIGHT, avanti e destra, per disegnare linee ad arco e semicirconferenze. Per le rotazioni in senso antiorario si usa l'istruzione RIGHT con valori negativi.

In questo progetto verranno utilizzate le istruzioni di colore per il riempimento delle forme, definendo il concetto di forma.

Sarà necessario utilizzare i concetti di matematica e geometria per il calcolo della circonferenza.

L'immagine verrà salvata in formato vettoriale SVG e verrà creata l'animazione del disegno in un formato condivisibile via web.

Parlare di girasoli permette anche di toccare altri argomenti quali:

- Successione di Fibonacci (34, 55) / (89, 144)
- Sezione aurea
- Numeri primi
- Eliotropismo

Inoltre si tratta di un fiore originario del Perù importato in Europa nel XVI secolo, per gli Incas dio del Sole e, questo, permette di trattare anche argomenti storico-geografici.

Il mito greco sull'eliotropismo, considerando che il girasole è arrivato in Europa molti secoli dopo, può aprire varie tematiche interessanti.

E come non parlare dei girasoli di Van Gogh?

01. SCOMPOSIZIONE DI UN GIRASOLE [girasole_scomposto_01.odt]

CIRCLE

La corolla presenta un disco esterno di fiori a forma lanceolata disposti a raggiera, i petali, e un disco interno di piccoli fiori disposti in spirali che producono i frutti (quelli che comunemente vengono chiamati semi).

Le dimensioni del disco interno daranno le proporzioni a tutto il disegno.

Il disegno del disco interno e la sua posizione all'interno dello spazio sono poche e semplici istruzioni.

Per disegnare un cerchio al centro del foglio è sufficiente richiamare la funzione di LibreLogo che disegna cerchi: **CIRCLE**.

È sufficiente indicare a **CIRCLE** il valore del diametro desiderato, per questo progetto verrà utilizzato il valore di **90**.

In un nuovo documento di *Writer* basta scrivere


```
CIRCLE 90
```

per avere una corolla con diametro di **90** punti tipografici al centro del foglio.

L'istruzione **CIRCLE** disegna la figura partendo dal centro della figura stessa.

In Introduzione – Primi Passi come utilizzare unità di misura differenti per i valori.

La corolla avrà un contorno nero e l'interno di colore verde.




02. ISTRUZIONE CIRCLE [girasole_scomposto_02.odt]

CIRCLE [asse1,
asse2,inizio,
fine, tipo]

L'istruzione **CIRCLE** disegna un cerchio, come promesso dal nome dell'istruzione stessa.

Passando un solo parametro si ottiene un cerchio con il diametro pari al valore indicato dal parametro.

```
CIRCLE 50
```



Senza penna si ottiene un cerchio senza il disegno del bordo.

```
PENUP
CIRCLE 50
```




L'istruzione **CIRCLE** accetta anche più di un parametro che permettono di disegnare porzioni di cerchio in modi diversi.

```
CIRCLE [50, 50, 3h, 12h, 1]
```

o

```
CIRCLE [50, 50, 90, 0, 1]
```

permettono di disegnare un settore di cerchio



I primi due parametri riguardano la dimensione degli assi, quello orizzontale e quello verticale, seguono il punto di inizio e quello di fine in senso orario, **0** o **12h** in alto rispetto al muso della tartaruga, **90** o **3h** destra, **180** o **6h** in basso, **270** o **9h** a sinistra. I punti di inizio e di fine possono essere indicati in gradi o con le ore dell'orologio, in questo caso occorre mettere il valore seguito dall'unità di misura "h" senza lasciare spazio.


L'ultimo parametro con il valore **1** indica di disegnare un settore.

```
CIRCLE [50, 50, 3h, 12h, 2]
```

o

```
CIRCLE [50, 50, 90, 0, 2]
```

permettono di disegnare un segmento di cerchio




L'unico valore diverso rispetto alle istruzioni precedenti è l'ultimo, il valore **2** indica di disegnare un segmento.

```
CIRCLE [50, 50, 3h, 12h, 3]
```

o

```
CIRCLE [50, 50, 90, 0, 3]
```

permettono di disegnare un arco



Il discriminante è sempre l'ultimo parametro che assume il valore 3.

In tutti gli esempi precedenti è stato utilizzato lo stesso valore per entrambi gli assi, se i valori fossero diversi si otterrebbe un'ellisse.

03. SOLO CONTORNO [girasole_scomposto_03.odt]

Il disco interno dovrà essere solo contorno, non nero. L'interno verrà "riempito" successivamente con una stilizzazione delle infiorescenze.

Nessuno degli esempi fatti per l'istruzione **CIRCLE** si traccia solo il contorno di un cerchio.

Per fare questo occorre utilizzare il colore.

Innanzitutto si stabilisce il colore della penna con l'istruzione **PENCOLOR** seguita dal parametro del colore scelto.

I colori in LibreLogo possono essere definiti in vari modi.

Con il nome, ad esempio "**olive**" per un colore marrone-verde

PENCOLOR

```
PENCOLOR "olive"
```

In questo caso il parametro va indicato tra apici "".

In Introduzione – Il Colore la tabella delle corrispondenze nome-colore.

Con il numero del colore, seguendo l'ordine della tabella, iniziando da 0, e volendo sempre il colore olive, si indica il valore numerico del parametro tra parentesi quadre, senza spazi tra le parentesi e il valore:

```
PENCOLOR [10]
```

Con la terna **RGB** definita dai valori (da 0 a 255) di rosso, verde, blu che compongono il colore desiderato, i valori che definiscono il solito colore olive sono **128** per la componente rosso, **128** per la componente verde e **0**, assenza, per la componente blu:

```
PENCOLOR [128,128,0]
```

Con il corrispondente esadecimale preceduto da **0x**:

```
PENCOLOR 0x808000
```

0xFFFFFFFF → Bianco; 0x000000 → Nero


FILLCOLOR

Per il riempimento delle forme si utilizza l'istruzione **FILLCOLOR**, si possono utilizzare gli stessi modi per indicare il parametro del colore desiderato.

Per creare un cerchio con il solo contorno si può utilizzare il valore **"invisible"** come parametro colore, in questo modo la forma non avrà alcun colore di riempimento.

Le impostazioni per il colore penna e di riempimento vanno scritte prima dell'istruzione **CIRCLE**.

```
PENCOLOR 0x808000
FILLCOLOR "invisible"
CIRCLE 90
```



Il disco interno farà da riferimento per tutto il disegno, il valore del diametro verrà utilizzato per definire le altre misure delle componenti del disegno.

Variabile

Per non dover ripetere il numero in tutte le parti dove servirà e per poter modificarne facilmente il valore in tutto il progetto, è possibile assegnare il valore ad una variabile. La variabile viene chiamata **d**, diametro.

Per assegnare un valore ad una variabile

```
d = 90
```

Assegnazione di variabile

Per utilizzare la variabile

```
CIRCLE d
```

Utilizzo come parametro

Il codice diventa

```
d = 90
PENCOLOR 0x808000
FILLCOLOR "invisible"
CIRCLE d
```

04. UN PETALO [girasole_scomposto_04.odt]

**FORWARD
RIGHT**

Per disegnare i petali esterni si useranno le sole funzioni **FORWARD** e **RIGHT**, avanti e destra, in questo modo si potranno creare delle linee ondulate e sarà possibile chiuderle per creare delle forme da riempire di colore.

REPEAT n [istruzioni]

Per creare delle curve occorre muovere in avanti di pochissimi punti 1 o 2 e subito ruotare di pochissimi gradi, ancora tra 1 e 2.

Queste due istruzioni andranno ripetute il numero di volte necessarie per disegnare una linea della lunghezza desiderata racchiudendole nell'istruzione **REPEAT numero volte [istruzioni]**.


Provare in modo empirico per trovare la curva voluta.
Ad esempio:

```
d = 90
PENCOLOR 0x808000
FILLCOLOR "invisible"

CIRCLE d

REPEAT d*(2/3) [
  FORWARD 2
  RIGHT 1.5
]
```

In questo modo viene disegnato un arco di cerchio partendo dal centro del disco.



Prima di continuare è opportuno fare in modo che i petali partano dalla circonferenza.

Una semicirconferenza può essere disegnata con la ripetizione di **FORWARD** e **RIGHT** oppure con l'istruzione **CIRCLE**.

Con la ripetizione di **FORWARD** e **RIGHT** il punto di partenza del disegno coincide con l'inizio del segmento e la tartaruga ne traccia il percorso.

Con **CIRCLE** la posizione della tartaruga è al centro della circonferenza e la tartaruga non si muove dal centro.

Un'altra cosa da tenere presente è la possibilità di creare forme chiuse con il primo metodo a differenza del secondo. Le forme chiuse possono essere colorate.

Nel caso del disegno del petalo il primo metodo mostra maggiori possibilità di utilizzo e flessibilità.

Per posizionarsi sulla circonferenza e non al centro occorre spostare la tartaruga in alto della lunghezza del raggio.

Per non lasciare traccia dello spostamento si utilizza la funzione **PENUP**.

Con **PENDOWN** si permette alla tartaruga di lasciare nuovamente la traccia.

Si traccia la circonferenza del disco interno, si solleva la penna, ci si sposta della lunghezza del raggio, si abbassa la penna e si ripete **FORWARD** e **RIGHT** per la lunghezza scelta.

In istruzioni:

PENUP


PENDOWN

```

d = 90
PENCOLOR 0x808000
FILLCOLOR "invisible"
CIRCLE d
PENUP
FORWARD d/2

PENDOWN

REPEAT d*(2/3) [
  FORWARD 2
  RIGHT 1.5
]
    
```



Come valore per lo spostamento avanti è stato scelto il valore di 2, per la rotazione il valore di 1.5.
Provare altri valori per vedere le differenze e scegliere i valori adeguati al proprio progetto sulla base dell'effetto che si vuole ottenere.

05. IL PETALO SI COMPLETA [girasole_scomposto_05.odt]

Una curva in senso opposto e la tartaruga torna indietro per accennare la punta del petalo.

Una ripetizione pari alla metà della precedente, una rotazione di quasi 180° per tornare indietro e un'altra ripetizione simile alla precedente.

```

d = 90
PENCOLOR 0x808000
FILLCOLOR "invisible"
CIRCLE d

PENUP
FORWARD d/2

PENDOWN
    
```

```


REPEAT 2*(d/3) [
  FORWARD 2
  RIGHT 1.5
]

REPEAT d/3 [
  FORWARD 2
  RIGHT -1.5
]

RIGHT 150

REPEAT d/3 [
  FORWARD 2
  RIGHT 1.5
]

```



Anche in questo caso è interessante provare valori differenti da quelli qui indicati per vedere differenti effetti.

06. TANTI PETALI [girasole_scomposto_06.odt]

Funzione

```

TO
  istruzioni
END

```

I petali sono disposti a raggiera lungo la circonferenza, si tratta di ripetere la stessa sequenza di istruzioni tante volte quanti sono i petali desiderati. La sequenza di istruzioni che disegnano un petalo si possono raggruppare in una funzione, in questo modo sarà sufficiente richiamare la funzione tante volte quanti sono i petali da disegnare.

Per definire una funzione le istruzioni vengono racchiuse tra le parole chiave **TO istruzioni END**.

In questo modo:

```
PENCOLOR 0x808000
```



```
FILLCOLOR "INVISIBLE"
```

```
d = 90
```

```
PENDOWN  
CIRCLE d  
PENUP
```

```
FORWARD d/2
```

```
TO DisegnaPetalò d
```

```
PENDOWN  
REPEAT 2*(d/3) [  
  FORWARD 2  
  RIGHT 1.5  
]  
REPEAT d/3 [  
  FORWARD 2  
  RIGHT -1.5  
]  
RIGHT 150  
REPEAT d/3 [  
  FORWARD 2  
  RIGHT 1.5  
]  
PENUP  
END
```

```
DisegnaPetalò d
```

Funzione
parametro/i

Per richiamare la funzione è sufficiente scriverne il nome e indicare il/i parametro/i come nell'ultima riga del codice sopra.

In questo caso la funzione si chiama **DisegnaPetalò** e richiede come parametro il valore di **d**, il diametro.

HOME

Fino a questo momento non sono state date indicazioni né per la posizione di partenza della tartaruga e nemmeno per la sua direzione. Ogni nuovo documento presenta la tartaruga al centro della pagina, in posizione **HOME**, orientata verso l'alto.

Per disporre i petali a raggiera un modo possibile è quello di disegnare il primo petalo, tornare al centro, ruotare, spostarsi sulla circonferenza, disegnare il secondo petalo e di nuovo tornare al centro e così via per tutti i petali.

Per posizionare sempre al centro la tartaruga, dopo il disegno di ogni petalo si può tornare alla posizione **HOME**.

Per un progetto più “flessibile” è possibile impostare una posizione di inizio come valore di una variabile chiamata **origine**. In questo modo sarà possibile definire il punto di origine del proprio girasole in una posizione differente dal centro.
Per fare questo si assegna una coppia di coordinate **x,y** ad una variabile che verrà chiamata **origine**.

Le variabili possono avere un qualsiasi nome si voglia, meglio se questo nome è significativo, i nomi delle variabili possono contenere lettere, numeri e underscore (il trattino basso, _), ma devono necessariamente iniziare con una lettera.

```
origine = [PAGESIZE[0]/2, PAGESIZE[1]/2]
```

PAGESIZE

Questa coppia di valori è l'equivalente di **HOME**, sfrutta i valori della variabile predefinita **PAGESIZE**. Il primo elemento di questa variabile, con indice **[0]**, rappresenta il valore della dimensione orizzontale, il secondo elemento, con indice **[1]**, quello della dimensione verticale.
Entrambi questi valori vengono divisi per 2, in questo modo si ottengono le coordinate **x,y** del centro del foglio, qualsiasi siano le dimensioni e l'orientamento del foglio.

POSITION

POSITION accetta due parametri, le coordinate **x,y** e consente di portare la tartaruga nella posizione desiderata.

```
POSITION origine
```

posiziona la tartaruga alle coordinate indicate da **origine**, in questo caso il centro del foglio.

Prima di disegnare ogni petalo si porterà la tartaruga alla posizione definita da **origine**.

Per ruotare i petali a raggiera si dovrà ruotare la tartaruga prima di ogni spostamento verso la circonferenza.

La rotazione sarà pari a **360** diviso il numero di petali desiderati.

Il numero di petali sarà assegnato ad una variabile chiamata **nP**, numero petali.

Le prime istruzioni sono per la penna e il colore di riempimento

```
PENCOLOR 0x808000
FILLCOLOR "INVISIBLE"
```

Seguono le assegnazioni di valori alle variabili: diametro, numero di petali, un indice contatore e le coordinate per il centro del disegno.

```
d = 90 ;diametro
nP = 11 ;numero petali
iP = 0 ;indice contatore petali
origine = [PAGESIZE[0]/2, PAGESIZE[1]/2]
```

La funzione **DisegnaPetalò**

```

TO DisegnaPetalò d
  PENDOWN
  REPEAT 2*(d/3) [
    FORWARD 2
    RIGHT 1.5
  ]
  REPEAT d/3 [
    FORWARD 2
    RIGHT -1.5
  ]

  RIGHT 150

  REPEAT d/3 [
    FORWARD 2
    RIGHT 1.5
  ]
  PENUP
END
  
```

All'inizio si posiziona la tartaruga alle coordinate di **origine**, si abbassa la penna, si traccia la circonferenza con diametro pari a **d**, si alza la penna.

```

POSITION origine
PENDOWN
CIRCLE d
PENUP
  
```

Il ciclo che ripete la funzione **DisegnaPetalò**, ogni volta posiziona al centro la tartaruga e la orienta per un angolo pari a **360** diviso il numero dei petali moltiplicato per il valore del contatore. All'inizio il contatore vale **0**, quindi la tartaruga sarà orientata verso l'alto, al termine del ciclo il contatore viene incrementato di **1**.

Si muove sulla circonferenza per prepararsi al disegno del petalo.

Viene richiamata la funzione **DisegnaPetalò**.

```



REPEAT nP [
  POSITION origine
  HEADING (360/nP)*iP
  FORWARD d/2

  DisegnaPetalò d


  iP = iP + 1
]
  
```

HEADING

L'istruzione **HEADING** è un'istruzione di stato della tartaruga e consente di ruotare la tartaruga verso i gradi indicati come parametro.




07. CORPO AL PETALO [girasole_scomposto_07.odt]



Per fare in modo che il petalo abbia più corpo e sia una forma che si può riempire di colore occorre fare in modo che il disegno del petalo diventi una forma chiusa.

Dopo il terzo e ultimo tratto della funzione DisegnaPetalò si procede con nuovi tratti per completare il disegno, per poter chiudere e riempire di colore la forma i tratti devono essere continui.

Una rotazione di quasi 180° per tornare indietro, ancora un tratto che quasi ricalca l'ultimo, di nuovo una rotazione di quasi 180° e 3 tratti che seguono e chiudono le due curve iniziali



Si inizia dal tratto rosso, poi il magenta, il blu, l'azzurro, il verde, il giallo e si finisce con il nero che chiude raccordandosi con il tratto rosso.

La funzione DisegnaPetalò cambia in questo modo, in rosso e grassetto le aggiunte:

```


T0 DisegnaPetalò d
PENDOWN
REPEAT 2*(d/3) [ ;tratto 1
FORWARD 2
    
```

```

    RIGHT 1.5
  ]
  REPEAT d/3 [ ;tratto 2
    FORWARD 2
    RIGHT -1.5
  ]
  RIGHT 150
  REPEAT d/3 [ ;tratto 3
    FORWARD 2
    RIGHT 1.5
  ]

  RIGHT 165
  REPEAT d/3 [ ;tratto 4
    FORWARD 2
    RIGHT -1
  ]
  RIGHT -150
  REPEAT d/3 [ ;tratto 5
    FORWARD 2
    RIGHT 1.5
  ]
  REPEAT d/3 [ ;tratto 6
    FORWARD 2
    RIGHT -1.5
  ]
  REPEAT d/3 [ ;tratto 7
    FORWARD 2
    RIGHT -1.9
  ]
  ]
  PENUP
  END

```



08. COLORARE I PETALI [girasole_scomposto_08.odt]

FILL

Per colorare la forma si utilizza l'istruzione **FILL**, il colore è quello impostato con **FILLCOLOR**. In questo momento **FILL** utilizza il colore "**invisible**", quello impostato per avere una circonferenza senza riempimento, occorre quindi impostare un nuovo colore per i petali con l'istruzione **FILLCOLOR** prima del ciclo di disegno dei petali

```
FILLCOLOR 0xeccd03
```

L'istruzione **FILL**, invece, può essere messa al termine del ciclo di disegno dei petali.

```
FILLCOLOR 0xeccd03
REPEAT 11 [
  POSITION origine
  HEADING (360/nP)*iP
  FORWARD d/2


  DisegnaPetalò d

  iP = iP + 1
]
FILL
```

In questo modo si otterrà il seguente risultato, il colore copre le linee di disegno.



Per avere le linee di disegno, l'istruzione **FILL** va posizionata prima di **PENUP** all'interno della funzione **DisegnaPetalò**, in questo modo



Come colore di contorno, in questo caso, è stato utilizzato il nero per rendere più evidente il disegno del contorno.

09. RIEMPIRE IL DISCO CENTRALE [girasole_scomposto_09.odt]

Il disco centrale del girasole è composto da un'infiorescenza di tanti piccoli fiori tubolosi a 5 denti. Sono disposti in numerose spirali in senso orario e antiorario. Di solito 34 in un senso e 55 nell'altro. Questi numeri sono due numeri consecutivi della successione di Fibonacci, i fiori più grandi ne hanno la coppia consecutiva 89 e 144. In questo progetto, per le dimensioni e la velocità di esecuzione verrà utilizzata la coppia precedente 13 e 21.

Le spirali verranno stilizzate con delle semicirconferenze a cui verrà dato "spessore" e chiuse come nel caso dei petali.

Quante ripetizioni servono per fare una semicirconferenza utilizzando le istruzioni **FORWARD** e **RIGHT**?

Di quanto occorre ruotare dopo ogni **FORWARD**?

La matematica e la geometria possono aiutare, si tratta, infatti, di calcolare la lunghezza della semicirconferenza.

La semicirconferenza si calcola dividendo per 2 il valore del diametro, che è nei dati di progetto, moltiplicato per π .

La lunghezza della semicirconferenza diviso il valore di quanto si vuole andare avanti ogni volta darà il numero delle ripetizioni necessarie.

Per quanto ruotare ogni volta è sufficiente dividere l'angolo piatto, 180° , per il numero di ripetizioni.

```

sc = INT (d*PI)/2 ;semicirconferenza
p = 2 ;valore per FORWARD
n = sc/p ;numero ripetizioni per la
          semicirconferenza
g = 180/n ;gradi di rotazione per la
           semicirconferenza
  
```

INT

INT è la funzione che permette di utilizzare solo la parte intera di un numero.

PI


PI restituisce il valore di π , pi greco.

```

sc = INT (d*PI)/2 ; semicirconferenza
p = 2 ; passi
n = sc/p ; numero ripetizioni per la
          semicirconferenza
g = 180/n ; gradi di rotazione per la
           semicirconferenza
  
```

```

REPEAT n [
  FORWARD p
  RIGHT g
]
  
```



Per dare spessore occorre ruotare di quasi 180° e creare una nuova semicirconferenza che verrà fermata poco prima, diminuendo il numero delle ripetizioni. La rotazione sarà minore, non una vera semicirconferenza, quindi. I gradi verranno calcolati dividendo 180 per il numero delle ripetizioni meno una certa quantità. La rotazione è con valore negativo.


La forma verrà chiusa grazie all'istruzione **FILL**. L'istruzione **FILL** permette di chiudere e colorare una forma.

Le istruzioni da aggiungere a quelle precedenti

```
RIGHT 175

REPEAT n-15 [
  FORWARD p
  RIGHT -180/(n-4)
]
FILL
```

FILLTRANSPAREN
CY



Il colore ha un valore di trasparenza impostato con **FILLTRANSPARENCY**.

Si possono racchiudere tutte le istruzioni in un'unica procedura denominata **DisegnaSemi**

```
FILLCOLOR 0xbcc553
FILLTRANSPARENCY 40

TO DisegnaSemi d
  sc = INT (d*PI)/2      ; semicirconferenza
  p = 2                  ; passi
  n = sc/p               ; numero ripetizioni per la
                        ; semicirconferenza
  g = 180/n              ; gradi di rotazione per la
                        ; semicirconferenza

  REPEAT n [
```



```

        FORWARD p
        RIGHT g
    ]

    RIGHT 175

    REPEAT n-15 [
        FORWARD p
        RIGHT -180/(n-4)
    ]
    FILL
    PENUP
END

DisegnaSemi d
  
```

10. LE SPIRALI DEI SEMI [girasole_scomposto_10.odt]

Anche nel caso delle spirali dei semi occorre partire sempre dalla posizione centrale e ruotare di 360 diviso il numero di spirali necessarie e richiamare la funzione **DisegnaSemi**.

Occorre passare un altro parametro alla funzione che indichi se spirale in senso orario o antiorario. Il valore di questo parametro potrà valere 1 o -1 e sarà il fattore di moltiplicazione dei valori di rotazione.

La funzione diventa

```

TO DisegnaSemi d rotazione
  sc = INT (d*PI)/2 ; semicirconferenza
  p = 2 ; passi
  n = sc/p ; numero ripetizioni per la
              semicirconferenza
  g = 180/n ; gradi di rotazione per la
              semicirconferenza

  PENDOWN
  REPEAT n [
    FORWARD p
    RIGHT g*rotazione
  ]

  RIGHT 175*rotazione

  REPEAT n-15 [
    FORWARD p
    RIGHT -180/(n-4)*rotazione
  ]
  FILL
  PENUP
END
  
```

Alla variabile **s** verrà assegnata la coppia di valori delle spirali

```
s = [13, 21] ; spirali (orarie, antiorarie)
```


Questa variabile verrà utilizzata nei due cicli che riempiranno il disco centrale rispettivamente con indice **[0]** e indice **[1]**

```
REPEAT s[0] [
  POSITION origine
  HEADING 0 + 360/s[0]*(REPCOUNT-1)
  DisegnaSemi d 1
]

REPEAT s[1] [
  POSITION origine
  HEADING 0 + 360/s[1]*(REPCOUNT-1)
  DisegnaSemi d -1
]
```

La funzione **DisegnaSemi** viene richiamata in due cicli successivi passando come parametro di rotazione prima il valore **1** e dopo **-1**. Si utilizza il contatore dei cicli **REPCOUNT** per ruotare la tartaruga e disegnare la semicirconferenza.

REPCOUNT



11. TUTTO INSIEME [girasole_scomposto_11.odt]

Si deve mettere tutto insieme.

```
PENCOLOR 0x808000
FILLCOLOR "INVISIBLE"

d = 90          ;diametro
nP = 19        ;numero petali
iP = 0         ;indice contatore petali
s = [13, 21]   ;spirali (orarie e antiorarie)
origine = [PAGESIZE[0]/2, PAGESIZE[1]/2]
```

Si assegna il colore alla penna e al colore di riempimento per il disco centrale, segue l'assegnazione di variabili per il diametro, il numero di petali, l'indice del contatore di petali, le spirali e l'origine.

```
POSITION origine

PENDOWN
CIRCLE d*2
```

Si posiziona la tartaruga alle coordinate stabilite da origine, si abbassa la penna e si traccia il disco centrale con un diametro doppio.

```
TO DisegnaSemi d rotazione
  sc = INT (d*PI)/2    ; semicirconferenza
  p = 2                ; passi
  n = sc/p             ; numero ripetizioni per la
                      ; semicirconferenza
  g = 180/n           ; gradi di rotazione per la
                      ; semicirconferenza

  PENDOWN
  REPEAT n [
    FORWARD p
    RIGHT g*rotazione
  ]

  RIGHT 175*rotazione

  REPEAT n-15 [
    FORWARD p
    RIGHT -180/(n-4)*rotazione
  ]
  FILL
  PENUP
END
```

La definizione della funzione **DisegnaSemi**

```

T0 DisegnaPetalò d
  PENDOWN
  REPEAT 2*(d/3) [ ;tratto 1
    FORWARD 2
    RIGHT 1.5
  ]
  REPEAT d/3 [ ;tratto 2
    FORWARD 2
    RIGHT -1.5
  ]
  RIGHT 150
  REPEAT d/3 [ ;tratto 3
    FORWARD 2
    RIGHT 1.5
  ]

  RIGHT 165
  REPEAT d/3 [ ;tratto 4
    FORWARD 2
    RIGHT -1
  ]
  RIGHT -150
  REPEAT d/3 [ ;tratto 5
    FORWARD 2
    RIGHT 1.5
  ]
  REPEAT d/3 [ ;tratto 6
    FORWARD 2
    RIGHT -1.5
  ]
  REPEAT d/3 [ ;tratto 7
    FORWARD 2
    RIGHT -1.9
  ]
  ]
  FILL
  PENUP
END

```

La funzione **DisegnaPetalò**

```

FILLCOLOR 0xbcc553 ;colore semi
FILLTRANSPARENCY 40

REPEAT s[0] [
  POSITION origine
  HEADING 0 + 360/s[0]*REPCOUNT
  DisegnaSemi d 1

```

```

]

REPEAT s[1] [
  POSITION origine
  HEADING 0 + 360/s[1]*REPCOUNT
  DisegnaSemi d -1
]

```

I cicli per il disegno del disco centrale


```

FILLCOLOR 0xeccd03 ;colore petali
REPEAT nP [
  POSITION origine
  HEADING (360/nP)*iP
  FORWARD d

  DisegnaPetalò d
  iP = iP + 1
]

```

Il ciclo per i petali esterni con il colore di riempimento giallo



12. GIRASOLE VETTORIALE [girasole_scomposto_12.odt – girasole_scomposto.svg]

Un'immagine vettoriale è un'immagine che mantiene le sue caratteristiche a qualsiasi scala. L'immagine, infatti, non perde risoluzione grafica. Sono immagini che lavorano in **XML**, un metalinguaggio per la definizione di linguaggi di markup, un metalinguaggio che permette lo scambio di informazioni, anche grafiche, attraverso il web.

Il formato **SVG** permette di trattare forme geometriche elementari e complesse, sono definite matematicamente e non per pixel come nelle immagini raster.

La qualità dell'immagine è pari alle possibilità offerte dal supporto (carta, monitor, smartphone,...).

PICTURE

Per trasformare le istruzioni in un'immagine SVG occorre utilizzare la funzione **PICTURE nome_file con estensione .svg [istruzioni]**.

L'immagine verrà salvata nella cartella dove è salvato il file dello script, è possibile indicare in *nome_file* un percorso completo sul proprio pc.

L'estensione di *nome_file* deve necessariamente essere **.svg**, ad esempio:

```
PICTURE "immagine1.svg" [
  istruzioni
]
```

Le istruzioni racchiuse in **PICTURE** non comprendono le proprie procedure/funzioni o la definizione delle variabili.

Per trasformare il girasole in immagine è sufficiente racchiudere i cicli che disegnano i petali e il disco centrale, nel modo seguente:

```
PICTURE "girasole_scomposto.svg" [
  PENCOLOR 0x808000
  FILLCOLOR "INVISIBLE"

  POSITION origine

  PENDOWN
  CIRCLE d*2

  FILLCOLOR 0xbcc553 ;colore semi
  FILLTRANSPARENCY 40

  REPEAT s[0] [
    POSITION origine
    HEADING 0 + 360/s[0]*REPCOUNT
    DisegnaSemi d 1
  ]

  REPEAT s[1] [
    POSITION origine
    HEADING 0 + 360/s[1]*REPCOUNT
```

```

        DisegnaSemi d -1
    ]

    FILLCOLOR 0xeccd03 ;colore petali
    REPEAT nP [
        POSITION origine
        HEADING (360/nP)*iP
        FORWARD d

        DisegnaPetalò d

        iP = iP + 1
    ]
  ]

```

In rosso e grassetto le istruzioni che erano poste in altre posizioni dello script e che vanno racchiuse in **PICTURE**. Il resto dello script resta identico e la funzione PICTURE va alla fine. Nella cartella dello script si può visualizzare il file "girasole_scomposto.svg" appena creato.

13. PETALO PER PETALO [girasole_scomposto_12.odt – girasole_animazione.svg]

CODICE
segnalibro

Il girasole si disegna petalo per petalo. Per creare un'animazione del fiore che si forma petalo per petalo si deve utilizzare la funzione **SLEEP**. **SLEEP** richiede come parametro un valore numerico che indica di quanti millisecondi si dovrà sospendere l'esecuzione. Mettendo **SLEEP 500** dopo ogni richiamo di azione di disegno non si avrà più un'immagine statica ma un'immagine che si compone. Al termine è stata inserita **SLEEP 10000**, dopo 10 secondi l'animazione riprenderà dall'inizio.

```

PICTURE "girasole_animazione.svg" [
    PENCOLOR 0x808000
    FILLCOLOR "INVISIBLE"

    POSITION origine

    PENDOWN
    CIRCLE d*2
    SLEEP 500

    FILLCOLOR 0xbcc553 ;colore semi
    FILLTRANSPARENCY 40

    REPEAT s[0] [
        POSITION origine

```

```
    HEADING 0 + 360/s[0]*REPCOUNT
    DisegnaSemi d 1
    SLEEP 500
]

REPEAT s[1] [
    POSITION origine
    HEADING 0 + 360/s[1]*REPCOUNT
    DisegnaSemi d -1
    SLEEP 500
]

FILLCOLOR 0xeccd03 ;colore petali
REPEAT nP [
    POSITION origine
    HEADING (360/nP)*iP
    FORWARD d

    DisegnaPetalò d
    SLEEP 500

    iP = iP + 1
]
SLEEP 10000
]
```


ISTRUZIONI

In corsivo il parametro (valore) necessario all'istruzione

CIRCLE *numero* - disegna un cerchio con diametro pari a quanto indicato da numero

FILL - Chiude e riempie le forme con il colore di riempimento indicato con *FILLCOLOR*

FILLCOLOR *colore* - Definisce il colore di riempimento delle forme

FILLTRANSPARENCY *numero* - Imposta la trasparenza di colore per la percentuale indicata dal numero

FORWARD *numero* - Si muove in avanti per il numero indicato

HEADING *numero* - Direzione della tartaruga in gradi, 0 → alto, 90 → destra, 180 → basso, 270 → sinistra

HOME - Porta la tartaruga al centro del foglio

INT - Restituisce la parte intera di un numero

PAGESIZE - Indica le dimensioni del foglio di lavoro, è un array con due elementi, *PAGESIZE[0]* → dimensione orizzontale, *PAGESIZE[1]* → dimensione verticale

PENCOLOR *colore* - Definisce il colore della penna

PENDOWN - Abbassa la penna per disegnare

PENUP - Alza la penna per non lasciare segni

PI - Restituisce il valore di π (pi greco)

PICTURE *nome-file.svg* [*istruzioni*] - Crea un'immagine SVG con il nome indicato

POSITION [*x*, *y*] - Posizione [*valore x*, *valore y*]

REPCOUNT - Contatore dei cicli come *REPEAT*

REPEAT *numero* [*istruzioni*] - Ripeti quante volte [*istruzioni da ripetere*]

RIGHT *numero* - Ruota in senso orario per il numero di gradi indicati



GIRASOLE SCOMPOSTO

SLEEP numero - Indica una pausa tra un'istruzione e l'altra in millisecondi

TO nome funzione (parametro/i) istruzioni **END** - Definisce le istruzioni come funzione